Laboratory 2: Difference Equations. Equilibrium Points. Periodic Points. Stability

```
> with(plots):
cobweb := proc(f, xmin, xmax, a0s := [], n := 0, init_gs := [])
  local i, j, x, a, l, gs, a0, color_list, idx;
  color_list := [green, purple, cyan, magenta, yellow, black];
  gs := [op(init_gs), plot([f(x), x], x=xmin..xmax, color=[red, 
blue], legend=["f(x)", "x"])]:
  idx := 1:
  for a0 in a0s do
    a[0] := a0:
    l[0] := [a[0], 0]:
    for i from 1 to n do
      a[i] := evalf(f(a[i-1])):
      l[2*i-1] := [a[i-1], a[i]]:
      l[2*i] := [a[i], a[i]]:
    end do:
    gs := [op(gs), plot(
      [l[j]$j=0..2*n],
      style=line,
      color=black,
      legend=sprintf("a0 = %g", a0),
      color=color_list[(idx-1) mod nops(color_list) + 1]
    )]:
    idx := idx + 1:
  end do:
  display(gs);
end:
```

1. Build a numerical solution for the following initial value problems. Plot your data to observe patterns in the solutions. Is there an equilibrium solution? Is it stable or unstable?

   (a) $a_{n+1} = -1.2a_n + 50$, $a_0 = 1000$;

   (b) $a_{n+1} = 0.8a_n - 100$, $a_0 = 500$;

   (c) $a_{n+1} = 0.8a_n - 100$, $a_0 = -500$;

   (d) $a_{n+1} = a_n - 100$, $a_0 = 1000$;

```
> restart;
> n := 30;
```

$$n := 30 \tag{1}$$

Exercise (1.a)

```
> a[0] := 1000:
for i from 1 to n do
  a[i] := -1.2 * a[i-1] + 50;
end do:
> plot([[j, a[j]]$j=0..n], style=point, symbol=circle);
```




```
> a_eq1 := solve(a = -1.2 * a + 50);
```

$$a\_eq1 := 22.72727273 \quad (2)$$

For this example, the rate of change is r = -1.2 and the recurrence relation has the solution a* = 22.72. Because |r| > 1, that's an unstable equilibrium point. In the graph, it can be seen that the sequence is increasing unbounded for even indices and is decreasing unbounded for odd indices.

Exercise (1.b)

```
> a[0] := 500:
  for i from 1 to n do
    a[i] := 0.8 * a[i-1] - 100;
  end do:
> plot([[j, a[j]]$j=0..n], style=point, symbol=circle);
```




```
> a_eq2 := solve(a = 0.8 * a - 100);
```

$$a\_eq2 := -500. \tag{3}$$

For this example, the rate of change is r = 0.8 and the recurrence relation has the solution a* = -500. Because |r| < 1, that's an stable equilibrium point. From the graph of this function, it seems that the sequence tends to the equilibrium point.

Exercise (1.c)

```
> a[0] := -500:
  for i from 1 to n do
    a[i] := 0.8 * a[i-1] - 100;
  end do:
> plot([[j, a[j]]$j=0..n], style=point, symbol=circle);
```



We have the same equation as before, but the starting point is exactly the equilibrium point, so the equation is a straight line.

Exercise (1.d)

```
> a[0] := 1000:
  for i from 1 to n do
    a[i] := a[i-1] - 100;
  end do:
> plot([[j, a[j]]$j=0..n], style=point, symbol=circle);
```




```
> a_eq3 := solve(a = a - 100);
```

$$a\_eq3 :=$$ **(4)**

For this example, the rate of change is r = 1 and the recurrence equation has no solution, so there is no equilibrium solution. We can see from the graph that it decreases unbounded.

2. For the following problems find the solution to the difference equation and the equilibrium value if one exists. Discuss the long-term behaviour of the solutions for various initial data. Clasify the equilibrium values as stable or unstable. Draw the Cobweb diagram for each equation with different initial starting points.

   (a) $a_{n+1} = -a_n + 2;$
   (b) $a_{n+1} = a_n + 2;$
   (c) $a_{n+1} = a_n + 3.2;$
   (d) $a_{n+1} = -3a_n + 4;$
   (e) $a_{n+1} = a_n^2 + 3a_n;$

```
> restart;
```

Exercise (2.a)

```
> f := x -> -x + 2;
```

$$f := x \rightarrow -x + 2 \quad \textbf{(5)}$$

```
> a_eq1 := solve(f(x) = x);
```

$$a\_eq1 := 1 \quad \textbf{(6)}$$

```
> deriv_mod := abs(D(f)(a_eq1));
```

$$deriv\_mod := 1 \quad \textbf{(7)}$$

```
> cobweb(
    f, 0, 2, [0.5, 0.7, 1, 1.2, 1.7], 10,
    [plot([[a_eq1, f(a_eq1)]], style=point, symbol=circle,
  symbolsize=20, legend="Equilibrium point")]
  );
```



In this case, the equilibrium points exists, but is unstable.

Exercise (2.b)

```
> f := x -> x + 2;
```

$$f := x \rightarrow x + 2 \quad \textbf{(8)}$$

```
> a_eq2 := solve(f(x) = x);
```

$$a\_eq2 := \quad \textbf{(9)}$$

```
> deriv_mod := abs(D(f)(a_eq2));
```
$$deriv\_mod := 1 \tag{10}$$

```
> cobweb(f, 0, 9, [0.5, 1, 1.5], 4);
```


In this case, the equilibrium point doesn't exist.

Exercise (2.c)

```
> f := x -> x + 3.2;
```
$$f := x \rightarrow x + 3.2 \tag{11}$$

```
> a_eq3 := solve(f(x) = x);
```
$$a\_eq3 := \tag{12}$$

```
> deriv_mod := abs(D(f)(a_eq3));
```
$$deriv\_mod := 1 \tag{13}$$

```
> cobweb(f, 0, 13, [0.5, 1, 1.5], 4);
```



In this case, the equilibrium point doesn't exist.

Exercise (2.d)

```
> f := x -> -3 * x + 4;
```

$$f := x \rightarrow -3\,x + 4 \quad \mathbf{(14)}$$

```
> a_eq4 := solve(f(x) = x);
```

$$a\_eq4 := 1 \quad \mathbf{(15)}$$

```
> deriv_mod := abs(D(f)(a_eq4));
```

$$deriv\_mod := 3 \quad \mathbf{(16)}$$

```
> cobweb(
    f, -150, 150, [0.5, 1, 1.5], 5,
    [plot([[a_eq4, f(a_eq4)]], style=point, symbol=circle,
  symbolsize=30, legend="Equilibrium point")]
  );
```



In this case, the equilibrium point exists, but is unstable.

Exercise (2.e)

```
> f := x -> x^2 + 3 * x;
```

$$f := x \rightarrow x^2 + 3\,x \quad \textbf{(17)}$$

```
> a_eq5 := solve(f(x) = x);
```

$$a\_eq5 := -2, 0 \quad \textbf{(18)}$$

```
> deriv_mod1 := abs(D(f)(a_eq5[1]));
```

$$deriv\_mod1 := 1 \quad \textbf{(19)}$$

```
> deriv_mod2 := abs(D(f)(a_eq5[2]));
```

$$deriv\_mod2 := 3 \quad \textbf{(20)}$$

```
> cobweb(
    f, -3.2, 0.4, [-2.5, -0.8, -1.4], 5,
    [plot([[a_eq5[1], f(a_eq5[1])]], style=point, symbol=circle,
  symbolsize=20, legend="Equilibrium Point")]
  );
```




```
> cobweb(
    f, -4, 10, [-3.05, 0.03], 5,
    [plot([[a_eq5[2], f(a_eq5[2])]], style=point, symbol=circle,
  symbolsize=30, legend="Equilibrium Point")]
  );
```



In this case, for x in [-3, 0] the equilibrium points exists and is unstable, and for x in R - [0, 3], the equilibrium point doesn't exist.

3. (Newton's Method of Computing the Square Root of a PositiveNumber)

The equation $x^2 = a$ can be written in the form $x = \frac{1}{2}(x + \frac{a}{x})$. This form leads to Newton's method

$$x_{n+1} = \frac{1}{2}(x_n + \frac{a}{x_n})$$

(a) Show that this difference equation has two equilibrium points $\sqrt{a}$ and $-\sqrt{a}$;

(b) Sketch a cobweb diagram for $a = 3$, $x_0 = 1$ and $x_0 = -1$.

```
> restart;
> a := 'a';
```

$$a := a \qquad (21)$$

```
> f := x -> 1/2 * (x + a / x);
```

$$f := x \rightarrow \frac{1}{2}\, x + \frac{1}{2}\, \frac{a}{x} \qquad (22)$$

Exercise (3.a)

```
> a_eq := solve(f(x) = x, x);
```

$$a\_eq := \sqrt{a}, -\sqrt{a} \quad (23)$$

Exercise (3.b)

```
> a := 3; f(x);
```

$$a := 3$$

$$\frac{1}{2}x + \frac{3}{2x} \quad (24)$$

```
> cobweb(
    f, 0.5, 5, [1, 3, 4], 5,
    [plot([[a_eq[1], f(a_eq[1])]], style=point, symbol=circle,
  symbolsize=20, legend="Equilibrium Point")]
  );
```




```
> cobweb(
    f, -0.5, -5, [-1, -3, -4], 5,
    [plot([[a_eq[2], f(a_eq[2])]], style=point, symbol=circle,
  symbolsize=20, legend="Equilibrium Point")]
  );
```



4. Let $f(x) = -\frac{1}{2}x^2 - x + \frac{1}{2}$ and the difference equation

$$x_{n+1} = f(x_n).$$

Show that 1 is an asymptotically stable 2-periodic point of f.

```
> restart;
> f := x -> -1/2 * x^2 - x + 1/2;
  a0 := 1;
```

$$f := x \to -\frac{1}{2}x^2 - x + \frac{1}{2}$$

$$a0 := 1 \quad \textbf{(25)}$$

Definition. Let b be in the domain of f. Then:

(i) b is called a periodic point of f if for some positive integer k, f^k (b) = b. Hence **a point is k-periodic** if it is a fixed point of f^k, that is, if it is an equilibrium point of the difference equation

$a_{n+1} = g(a_n)$, where $g = f^k$.

The periodic orbit of b, O(b) = {b, f(b),f^2 (b), . . . , f^{k-1} (b)}, is often called a **k-cycle**.

(ii) b is called **eventually k-periodic** if for some positive integer m, f^m (b) is a k-periodic point. In other words, b is eventually k-periodic if

f^{m+k} (b) = f^m (b).

```
> f(f(a0)) = a0; f(a0) = a0;
```

$$1 = 1$$
$$-1 = 1 \tag{26}$$

Because f(f(a0)) is equal to a0 and f(a0) is not equal to a0, it means that a0 is a 2-periodic point. We need to check its stability.

Theorem. (Stability of k-cycle)

Let O(b) = {b=a_0, f(b),f^2 (b), . . . , f^{k-1} (b)} be a k-cycle of a continuously differentiable function f. Then the following statements hold:

(i) The k-cycle O(b) is asymptotically stable if

$$| f '(a_0)\ f '(a_1) \dots f '(a_{k-1})| < 1$$

(ii) The k-cycle O(b) is unstable if

$$| f '(a_0)\ f '(a_1) \dots f '(a_{k-1})| > 1$$

```
> a := [a0, f(a0)];
```

$$a := [1, -1] \tag{27}$$

```
> k_cycle = [D(f)(a[1]), D(f)(a[2])];
```

$$k\_cycle = [-2, 0] \tag{28}$$

Because a0 is a 2-periodic point of f and the product of the elements from the 2-cycle is 0, it means that a0 is an asymptotically stable 2-periodic point of f.

5. Find the solution for the following difference equations:

(a) $x_{n+2} - 5x_{n+1} + 6x_n = 0,\ x_0 = 1,\ x_1 = 1;$

(b) $y_{n+2} - 3y_{n+1} + 2y_n = 2n^2 + 6n,\ y_0 = 1,\ y_1 = 2;$

(c) $y_{n+2} + 3y_n + 2y_n = 3^n(2n^2 + 4n),\ y_0 = 2,\ y_1 = 1;$

```
> restart;
```

Exercise (5.a)

```
> eq1 := x(n+2) - 5 * x(n+1) + 6 * x(n) = 0;
```

$$eq1 := x(n+2) - 5\,x(n+1) + 6\,x(n) = 0 \tag{29}$$

```
> x(0) := 1; x(1) := 1;
```

$$x(0) := 1$$
$$x(1) := 1 \quad (30)$$

```
> sol1 := rsolve(eq1, x(n));
```

$$sol1 := -3^n + 2\, 2^n \quad (31)$$

```
> sol1 := simplify(sol1);
```

$$sol1 := -3^n + 2^{n+1} \quad (32)$$

Exercise (5.b)

```
> eq2 := y(n+2) - 3 * y(n+1) + 2 * y(n) = 2 * n^2 + 6 * n;
```

$$eq2 := y(n+2) - 3\, y(n+1) + 2\, y(n) = 2\, n^2 + 6\, n \quad (33)$$

```
> y(0) := 1; y(1) := 2;
```

$$y(0) := 1$$
$$y(1) := 2 \quad (34)$$

```
> sol2 := rsolve(eq2, y(n));
```

$$sol2 := 13\, 2^n - 4\,(n+1)\left(\frac{1}{2}\, n + 1\right)\left(\frac{1}{3}\, n + 1\right) - 8 \quad (35)$$

```
> sol2 := simplify(sol2);
```

$$sol2 := -\frac{2}{3}\, n^3 - 4\, n^2 + 13\, 2^n - \frac{22}{3}\, n - 12 \quad (36)$$

Exercise (5.c)

```
> eq3 := y(n+2) + 3 * y(n+1) + 2 * y(n) = 3^n * (2 * n^2 + 4 * n);
```

$$eq3 := y(n+2) + 3\, y(n+1) + 2\, y(n) = 3^n\left(2\, n^2 + 4\, n\right) \quad (37)$$

```
> y(0) := 1; y(1) := 2;
```

$$y(0) := 1$$
$$y(1) := 2 \quad (38)$$

```
> sol3 := rsolve(eq3, y(n));
```

$$sol3 := \frac{73}{16}\,(-1)^n - \frac{429}{125}\,(-2)^n + \frac{79}{2000}\, 3^n + \left(-\frac{37}{100}\, n - \frac{37}{100}\right) 3^n + \frac{1}{5}\,(n+1)\left(\frac{1}{2}\, n + 1\right) 3^n \quad (39)$$

```
> sol3 := simplify(sol3);
```

$$sol3 := \frac{1}{10}\, 3^n\, n^2 + \frac{429}{125}\,(-1)^{n+1}\, 2^n - \frac{7}{100}\, 3^n\, n + \frac{73}{16}\,(-1)^n - \frac{261}{2000}\, 3^n \quad (40)$$

6. Consider the simple interest formula $S_n = (1 + np)S_0$ and the compound interest formula $S_n = (1 + p/r)^n S_0$. There are three options to earn interest. Company A offers simple interest at a rate of 6%. Company B offers compound interest at a 4% rate with a conversion period of one month. Company C offers compound interest at a 4% rate with a conversion period of three months.

   (a) Calculate for the three cases the amount on deposit after 5, 10, 15, and 20 years for any principal S_0.

   (b) Which interest offer maximizes the amount on deposit after 5, 10, 15,and 20 years?

```
> restart;
> simple_rate := p -> n -> evalf((1 + n * p) * S0);
```

$$simple\_rate := p \rightarrow n \rightarrow evalf((1 + n * p) * S0) \tag{41}$$

```
> compound_rate := (p, r) -> n -> evalf((1 + p / r) ^ (n * r) * S0)
  ;
```

$$compound\_rate := (p, r) \rightarrow n \rightarrow evalf((1 + p / r)^\wedge(n * r) * S0) \tag{42}$$

Exercise (6.a)

```
> f1 := simple_rate(6/100);
```

$$f1 := n \rightarrow evalf\left(\left(1 + n \frac{3}{50}\right) S0\right) \tag{43}$$

```
> f2 := compound_rate(4/100, 12/1);
```

$$f2 := n \rightarrow evalf\left(\left(1 + \frac{1}{25 \cdot 12}\right)^{n \cdot 12} S0\right) \tag{44}$$

```
> f3 := compound_rate(4/100, 12/3);
```

$$f3 := n \rightarrow evalf\left(\left(1 + \frac{1}{25 \cdot 4}\right)^{n \cdot 4} S0\right) \tag{45}$$

```
> S0 := 1;
```

$$S0 := 1 \tag{46}$$

```
> years := [5, 10, 15, 20];
```

$$years := [5, 10, 15, 20] \tag{47}$$

```
> plot([f1(x), f2(x), f3(x)], x=0..30, color=[red, blue, green],
  legend=[f1, f2, f3]);
```




```
> for year in years do
    printf("Year %d -> Company A: %g * S0 | Company B: %g * S0 |
  Company C: %g * S0\n", year, f1(year), f2(year), f3(year));
  end do;
Year 5 -> Company A: 1.3 * S0 | Company B: 1.221 * S0 | Company
C: 1.22019 * S0
Year 10 -> Company A: 1.6 * S0 | Company B: 1.49083 * S0 |
Company C: 1.48886 * S0
Year 15 -> Company A: 1.9 * S0 | Company B: 1.8203 * S0 |
Company C: 1.8167 * S0
Year 20 -> Company A: 2.2 * S0 | Company B: 2.22258 * S0 |
Company C: 2.21672 * S0
```

Exercise (6.b)

```
> interf1f2 := evalf(solve(f1(x) = f2(x)));
  interf1f3 := evalf(solve(f1(x) = f3(x)));
  interf2f3 := evalf(solve(f2(x) = f3(x)));
```

$$interf1f2 := -2.504164606\ 10^{-1812}, 19.17300115$$

$$interf1f3 := -8.374930901\ 10^{-144}, 19.38447971$$

$$interf2f3 := 0. \tag{48}$$

```
> with(plots):
  g1 := plot([f1(x), f2(x), f3(x)], x=0..30, color=[red, blue,
  green], legend=[f1, f2, f3]):
  g2 := plot(
    [[interf1f2[2], f1(interf1f2[2])]], style=point, symbol=circle,
  color=orange, symbolsize=20, legend="f1 ∩ f2"
  ):
  g3 := plot(
    [[interf1f3[2], f1(interf1f3[2])]], style=point, symbol=circle,
  color=purple, symbolsize=20, legend="f1 ∩ f3"
  ):
  g4 := plot(
    [[years[1], f1(years[1])], [years[2], f1(years[2])], [years[3],
  f1(years[3])], [years[4], f2(years[4])]],
    style=point, symbol=box, color=magenta, symbolsize=15, legend=
  "Highest values in given years"
  ):
  display(g1, g2, g3, g4);
```



We can observe that for year 5, 10, and 15, company A maximizes the amount on deposit, but for year 20 both company B and C are better than A.
In all the cases it seem like company B offers more than company C, as their single intersection is in 0, and for at least 1 positive number company B offers more than company C.

7. The loan on a house is \$200,000.

(a) Calculate the monthly repayment needed to have the loan repaid after 30 years. The interest rate is 5%.

(b) Calculate the total amount paid back on the loan.

```
> restart;
> loan := 200000; total_months := 30 * 12; monthly_rate := 5/100 /
  12;
```

$$loan := 200000$$
$$total\_months := 360$$
$$monthly\_rate := \frac{1}{240} \tag{49}$$

Exercise (7.a)

The value of a monthly payment made at month *k* is:

monthly_value_k = monthly_payment / (1 + monthly_rate)^k

These should be added up to the initial loan to be able to repay after 30 years, so:

loan = sum from *k=0* to *total_months-1* of *monthly_value_k =*
= sum from *k=0* to *total_months-1* of *monthly_payment / (1 + monthly_rate)^k =>*
=> monthly_payment = loan / (sum from *k=0* to *total_months-1* of *1 / (1 + monthly_rate)^k)*

```
> monthly_payment := evalf(loan / sum(1 / (1 + monthly_rate)^k, k=
  0..total_months));
```

$$monthly\_payment := 1067.910472 \tag{50}$$

Exercise (7.b)

The total payment can be found by multiplying the monthly payment by the number of total months.

```
> total_payment := monthly_payment * total_months;
  interest := total_payment - loan;
```

$$total\_payment := 3.844477699\ 10^{5}$$
$$interest := 1.844477699\ 10^{5} \tag{51}$$

This calculation highlights how interest can significantly increase the cost of borrowing over a long repayment period like 30 years, even at a relatively low annual interest rate of 5%.