

Assignment 1

1. Find the solution for the following difference equations:

(a) $x_{n+1} = \left(\frac{n+1}{n+2}\right)^2 \cdot x_n + \frac{1}{n+2}, \quad x_0 = 1;$

(b) $x_{n+3} - 4 \cdot x_{n+2} + x_{n+1} + 6 \cdot x_n = 60 \cdot 4^n, \quad x_0 = 2, \quad x_1 = 12, \quad x_2 = 12;$

(c) $x_{n+1} = \frac{2 \cdot x_n}{1 + 4 \cdot x_n}, \quad x_0 = 1$ (Hint: use substitution $x_n = \frac{1}{y_n}$)

Exercise (1.a)

```
> restart;
```

```
> eq_a := x(n+1) = ((n + 1) / (n + 2))^2 * x(n) + 1 / (n + 2);  
x(0) := 1;
```

$$eq_a := x(n+1) = \frac{(n+1)^2 x(n)}{(n+2)^2} + \frac{1}{n+2}$$

$$x(0) := 1$$

(1)

```
> sol_a := rsolve(eq_a, x(n));
```

$$sol_a := \frac{1}{2} \frac{n^2 + 3n + 4}{(n+1)^2}$$

(2)

Exercise (1.b)

```
> eq_b := x(n+3) - 4 * x(n+2) + x(n+1) + 6 * x(n) = 60 * 4^n;  
x(0) := 2; x(1) := 12; x(2) := 12;
```

$$eq_b := x(n+3) - 4x(n+2) + x(n+1) + 6x(n) = 60 \cdot 4^n$$

$$x(0) := 2$$

$$x(1) := 12$$

$$x(2) := 12$$

(3)

```
> sol_b := rsolve(eq_b, x(n));
```

$$sol_b := -4(-1)^n - 16 \cdot 3^n + 16 \cdot 2^n + 6 \cdot 4^n$$

(4)

Exercise (1.c)

```
> restart;
```

```
> eq_c := x(n+1) = (2 * x(n)) / (1 + 4 * x(n));  
x(0) := 1;
```

$$eq_c := x(n+1) = \frac{2x(n)}{1 + 4x(n)}$$

$$x(0) := 1$$

(5)

```
> sol_c := rsolve(eq_c, x(n));
```

$$sol_c := \text{rsolve}\left(x(n+1) = \frac{2x(n)}{1 + 4x(n)}, x(n)\right)$$

(6)

```
> eq_c2 := subs(x(n) = 1/y(n), x(n+1) = 1/y(n+1), eq_c);
```

```
y(0) := 1/x(0);
```

$$eq_c2 := \frac{1}{y(n+1)} = \frac{2}{y(n) \left(1 + \frac{4}{y(n)}\right)}$$

$$y(0) := 1$$

(7)

```
> sol_c2 := rsolve(eq_c2, y(n));
```

$$sol_c2 := -3 \left(\frac{1}{2}\right)^n + 4$$

(8)

```
> sol_c := simplify(1/sol_c2);
```

$$sol_c := -\frac{1}{3 \cdot 2^{-n} - 4}$$

(9)

2. Let us consider the difference equation:

$$x_{n+1} = \frac{x_n^2 + 7}{2x_n}.$$

- (a) Find the equilibrium points and study their stability.
- (b) Make some numerical simulations.

```
> restart;
```

```
> f := x -> (x^2 + 7) / (2 * x);
```

$$f := x \rightarrow \frac{1}{2} \frac{x^2 + 7}{x}$$

(10)

Exercise (2.a)

The equilibrium points are given by the solutions of the equation $f(x) = x$.

```
> sols := solve(f(x) = x);
```

$$sols := -\sqrt{7}, \sqrt{7}$$

(11)

To find their type of stability, we should compute $|f'(x)|$ and compare it with 1.

```
> df := D(f);
```

$$df := x \rightarrow 1 - \frac{1}{2} \frac{x^2 + 7}{x^2}$$

(12)

```
> abs(df(sols[1]));
```

$$0$$

(13)

Because $|f'(\text{first solution})|$ is 0 which is less than 1, it means that the first solution is a locally stable equilibrium point.

```
> abs(df(sols[2]));
```

$$0$$

(14)



Because $|f'(\text{second solution})|$ is 0 which is less than 1, it means that the second solution is a locally stable equilibrium point.

Exercise (2.b)

We can see that $f(x)$ has a vertical asymptote in $x=0$. We will make 2 numerical simulations where we will plot the cobweb diagram, one for $x \in (-12, -0.2)$ and one for $x \in (0.2, 12)$.


```
> with(plots):
cobweb := proc(f, xmin, xmax, a0s := [], n := 0, init_gs := [])
  local i, j, x, a, l, gs, a0, color_list, idx;
  color_list := [green, purple, cyan, magenta, yellow, black];
  gs := [op(init_gs), plot([f(x), x], x=xmin..xmax, color=[red,
blue], legend=["f(x)", "x"])] :
  idx := 1:
  for a0 in a0s do
    a[0] := a0:
    l[0] := [a[0], 0]:
    for i from 1 to n do
      a[i] := evalf(f(a[i-1])):
      l[2*i-1] := [a[i-1], a[i]]:
      l[2*i] := [a[i], a[i]]:
    end do:
    gs := [op(gs), plot(
      [l[j]$j=0..2*n],
      style=line,
      color=black,
      legend=sprintf("a0 = %g", a0),
      color=color_list[(idx-1) mod nops(color_list) + 1]
    )]:
    idx := idx + 1:
  end do:
  display(gs);
end:

> cobweb(
  f, -12, -0.2, [-10, -7, -3, -2, -1, -0.6, -0.3], 10,
  [plot([sols[1], f(sols[1])], style=point, symbol=circle,
legend="Equilibrium point", symbolsize=20)]
);
```



For the negative numbers simulations, it can be clearly seen that $f^n(x)$ tends to the negative equilibrium point, so that confirms that our point is a locally stable equilibrium point.

```
> cobweb(
  f, 0.2, 12, [0.3, 0.6, 1, 2, 3, 7, 10], 10,
  [plot([sols[2], f(sols[2])], style=point, symbol=circle,
    legend="Equilibrium point", symbolsize=20)]
);
```



For the positive numbers simulations, it also can be clearly seen that $f^n(x)$ tends to the positive equilibrium point, so that confirms that this point is also a locally stable equilibrium point.

3. Let us consider the difference equation:

$$x_{n+1} = x_n^2 - 3.$$

(a) Find the 2-periodic cycle and study its stability.

(b) Make numerical simulation.

```
> restart;
> f := x -> x^2 - 3;
```

$$f := x \rightarrow x^2 - 3$$

(15)

Exercise (3.a)

```
> sols_2 := solve(f(f(x)) = x);
```

(16)

$$sols_2 := -2, 1, \frac{1}{2} - \frac{1}{2} \sqrt{13}, \frac{1}{2} + \frac{1}{2} \sqrt{13} \quad (16)$$

```
> sols_1 := solve(f(x) = x);
```

$$sols_1 := \frac{1}{2} + \frac{1}{2} \sqrt{13}, \frac{1}{2} - \frac{1}{2} \sqrt{13} \quad (17)$$

We can see that out of the 4 solutions of the equation $f(f(x)) = x$, 2 of them are also solution of the equation $f(x) = x$, so they are not apart of a 2-periodic cycle.

So the 2-periodic cycle is composed by the first 2 solutions of the equation $f(f(x)) = x$. We verify that:

```
> sols := [sols_2[1], sols_2[2]];
```

$$sols := [-2, 1] \quad (18)$$

```
> sols; [f(sols[1]), f(sols[2])];
```

$$[-2, 1]$$

$$[1, -2] \quad (19)$$

We can see that these are composing the 2-periodic cycle. We will consider $b=-2$ our 2-periodic point, and we will check its stability (the equivalent holds for its pair).

To determine the stability of the 2-periodic point, we need to compute $|f'(sols1) * f'(sols2)|$ and compare it to 1.

```
> df := D(f);
```

$$df := x \rightarrow 2x \quad (20)$$

```
> abs(df(sols[1]) * df(sols[2]));
```

$$8$$


$$(21)$$

We can see that $|f'(sols1) * f'(sols2)|$ is 8, so the 2-periodic cycle is unstable.


Exercise (3.b)

To plot the numerical simulations, we firstly run the cobweb procedure definition block from exercise 2, then we will run the following blocks.

```
> cobweb(f, -4, 4, [sols[1] - 0.107, sols[1] + 0.001, sols[1]], 8);
```



```
> cobweb(f, -4, 4, [sols[2] - 0.003, sols[2] + 0.008, sols[2]], 8);
```



4. Consider the simple interest formula $S_n = (1 + np)S_0$ and the compound interest formula $S_n = (1 + p/r)^n S_0$. There are two options to earn interest. Company A offers simple interest at a rate of 4%. Company B offers compound interest at a 3% rate with a conversion period of one month.

(a) Calculate for the both cases the amount on deposit after 5, 10, 15, and 20 years for principal $S_0 = 1000$.

(b) Which interest offer maximizes the amount on deposit after 5, 10, 15 and 20 years?

```
> restart;
> simple_rate := p -> n -> evalf((1 + n * p) * S0);
                                simple_rate := p -> n -> evalf((1 + n * p) * S0) (22)
> compound_rate := (p, r) -> n -> evalf((1 + p / r) ^ (n * r) * S0)
;
                                compound_rate := (p, r) -> n -> evalf((1 + p / r) ^ (n * r) * S0) (23)
> f1 := simple_rate(4/100);
                                f1 := n -> evalf((1 + n * 1/25) * S0) (24)
> f2 := compound_rate(3/100, 12/1);
```


$$f2 := n \rightarrow evalf\left(\left(1 + \frac{3}{100 \cdot 12}\right)^{n \cdot 12} S0\right) \quad (25)$$

Exercise (4.a)

```
> S0 := 1000;
   years := [5, 10, 15, 20];
                                     S0 := 1000
                                     years := [5, 10, 15, 20] (26)
```


```
> for year in years do
   printf("Year %d | Company A: %g | Company B: %g\n", year, f1
   (year), f2(year));
end do;
```

```
Year 5 | Company A: 1200 | Company B: 1161.62
Year 10 | Company A: 1400 | Company B: 1349.35
Year 15 | Company A: 1600 | Company B: 1567.43
Year 20 | Company A: 1800 | Company B: 1820.75
```

Exercise (4.b)

We will plot the values for years 0..30.

```
> plot([f1(x), f2(x)], x=0..30, color=[red, blue], legend=["Company
A", "Company B"]);
```




It can be seen that the f_1 is bigger than f_2 except some year, where they intersect, than f_2 is bigger than f_1 . We will find that point:

```
> sols := evalf(solve(f1(x) = f2(x)));
sols := 2.669998613 10-792, 18.43930863 (27)
```

After the computations, it seems that f_1 and f_2 intersects in 2 points. We will plot those points:

```
> with(plots):
g1 := plot([f1(x), f2(x)], x=0..30, color=[red, blue], legend=
["Company A", "Company B"]):
g2 := plot(
[[sols[1], f1(sols[1])], [sols[2], f1(sols[2])]], style=point,
symbol=circle, symbolsize=20, legend="f1 ∩ f2"
):
g3 := plot(
[[years[1], f1(years[1])], [years[2], f1(years[2])], [years[3],
f1(years[3])], [years[4], f2(years[4])]],
style=point, symbol=box, symbolsize=15, color=magenta, legend=
"Highest values in the given years"
):
display(g1, g2, g3);
```



It can be seen from the graph that on the interval $[0, \text{sols}[1])$, $f2$ is bigger than $f1$ (but $\text{sols}[1]$ is really close to 0 so its negligible), for the interval $(\text{sols}[1], \text{sols}[2])$ $f1$ is bigger than $f2$, then for the interval $(\text{sols}[2], \infty)$ $f2$ is bigger than $f1$ (as $f2$ grows exponentially and $f1$ grows linearly).

Since years 5, 10, and 15 fall within the interval $(\text{sols}[1], \text{sols}[2])$, and year 20 lies in the interval $(\text{sols}[2], \infty)$, this indicates that Company A provides the highest deposit amount for the first three time points, while Company B offers the maximum deposit amount for the final year.